Math 174
Calculus II

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 23

Polar Coordinates

Initializations

```
> restart;
  with(plots):
>
```

23.1 Polar Coordinates

Instead of measuring the position of a point P by its x and y-coordinates, we can use the distance r from P to the origin and the angle θ between the line segment OP and the positive x-axis. The r and θ are called polar coordinates of the point P.

- The conversion formulas from polar coordinates (r, θ), to Cartesian coordinates (x, y) are

$$x = r\cos\theta \quad \text{and} \quad y = r\sin\theta$$

- The conversion formulas from Cartesian coordinates (x, y), to polar coordinates (r, θ) are

$$r = \sqrt{x^2 + y^2} \quad \text{and} \quad \theta = \arctan(y, x)$$

Note 1: Polar coordinates are not unique, every point (x, y) has infinitely many polar coordinates (r, θ).
Note 2: Unlike its one argument counterpart **arctan**(s), the two argument arctangent function **arctan**(y, x), has a range $(-\pi, \pi]$. Hence, it allows us to quickly find a θ coordinate for any point (x, y) in the plane.

Mathematical details will be provided in class.

Examples

Example 23.1.1

Find the Cartesian coordinates of the point $(r, \theta) = \left(5, \frac{\pi}{3}\right)$.

Solution
Use the formulas for conversion from polar to Cartesian coordinates.

```
> ans:=(x,y)=(5*cos(Pi/3), 5*sin(Pi/3));
```

$$ans := (x, y) = \left(\frac{5}{2}, \frac{5}{2}\sqrt{3}\right) \qquad \textbf{(2.1.1.1)}$$

```
>
```

Example 23.1.2

Find polar coordinates of the point $(x, y) = \left(7, \frac{7}{\sqrt{3}}\right)$.

Solution

Use the formulas for conversion from Cartesian to Polar coordinates.

```
> ans:=(r, theta)=(sqrt(7^2+(7/sqrt(3))^2), arctan(7/sqrt(3), 7));
```

$$ans := (r, \theta) = \left(\frac{14}{3}\sqrt{3}, \frac{1}{6}\pi\right) \quad \textbf{(2.1.2.1)}$$

```
>
```

Example 23.1.3

Give the polar equation of the line $x + 2y = 4$.

Solution

It is easy to convert a Cartesian equation in polar coordinates. One just needs to substitute the equations

$$x = r\cos\theta \quad \text{and} \quad y = r\sin\theta$$

into the Cartesian expression and, if possible, solve the result for *r*.

```
> e1:=x+2*y=4;
```

$$e1 := x + 2y = 4 \quad \textbf{(2.1.3.1)}$$

```
> e2:=subs({x=r*cos(theta), y=r*sin(theta)}, e1);
```

$$e2 := r\cos(\theta) + 2r\sin(\theta) = 4 \quad \textbf{(2.1.3.2)}$$

```
> ans:=isolate(e2, r);
```

$$ans := r = \frac{4}{\cos(\theta) + 2\sin(\theta)} \quad \textbf{(2.1.3.3)}$$

```
>
```

Example 23.1.4

Plot the cardioid

$$r = f(\theta) = 1 - 2\cos\theta$$

i) Using the parametric plot routine.

ii) Using the polarplot routine.

Solution

i) Plot the cardioid $r = f(\theta) = 1 - 2\cos\theta$, using the parametric plot routine.

Feed the parametric equations

$$x = f(\theta)\cos\theta \quad \text{and} \quad y = f(\theta)\sin\theta$$

into the parametric plot routine.

```
> f:=theta->1-2*cos(theta);
plot([f(theta)*cos(theta), f(theta)*sin(theta), theta=0..2*Pi], color=blue, scaling=constrained);
```

$$f := \theta \rightarrow 1 - 2\cos(\theta)$$



ii) Plot the cardioid $r = f(\theta) = 1 - 2\cos\theta$, using the polarplot routine.
Feed the function $f(\theta)$ into the **polarplot** routine.

```
> polarplot(f(theta), theta=0..2*Pi, color=red, scaling=
  constrained);
```




```
>
```

Example 23.1.5

Plot the rose

$$r = \sin 3\theta$$

Solution

Since the coefficient 3 is odd, we only need to extend our plot over the interval $[0, \pi]$.

```
> f:=theta->sin(3*theta);
  polarplot(f(theta), theta=0..Pi, scaling=constrained);
```

$$f := \theta \rightarrow \sin(3\,\theta)$$



>

Example 23.1.6

Find the equation of the tangent line at the point corresponding to $\theta = \frac{5\pi}{4}$ to the cardioid

$$\boldsymbol{r = 3 - 3\sin\theta}$$

Plot the cardioid and the tangent line in one picture.

Solution

First we compute the slope of the tangent line. Let $f(\theta) = 3 - 3\sin\theta$.

$$m = \left.\frac{dy}{dx}\right|_{\theta = \frac{5\pi}{4}} = \left.\frac{\frac{dy}{d\theta}}{\frac{dx}{d\theta}}\right|_{\theta = \frac{5\pi}{4}} = \left.\frac{\frac{d}{d\theta}[f(\theta)\sin\theta]}{\frac{d}{d\theta}[f(\theta)\cos\theta]}\right|_{\theta = \frac{5\pi}{4}}$$

```
> f:=theta->3-3*sin(theta);
```

$$f := \theta \rightarrow 3 - 3\sin(\theta) \quad \textbf{(2.1.6.1)}$$

```
> dy_dx:=diff(f(theta)*sin(theta), theta)/diff(f(theta)*cos
(theta), theta);
```

$$dy\_dx := \frac{-3\cos(\theta)\sin(\theta) + (3 - 3\sin(\theta))\cos(\theta)}{-3\cos(\theta)^2 - (3 - 3\sin(\theta))\sin(\theta)} \quad \textbf{(2.1.6.2)}$$

```
> m:=subs(theta=5*Pi/4, dy_dx);
```

$$m := \frac{-3\cos\left(\frac{5}{4}\pi\right)\sin\left(\frac{5}{4}\pi\right) + \left(3 - 3\sin\left(\frac{5}{4}\pi\right)\right)\cos\left(\frac{5}{4}\pi\right)}{-3\cos\left(\frac{5}{4}\pi\right)^2 - \left(3 - 3\sin\left(\frac{5}{4}\pi\right)\right)\sin\left(\frac{5}{4}\pi\right)} \quad \textbf{(2.1.6.3)}$$

```
> m:=expand(m);
```

$$m := -\sqrt{2} - 1 \quad \textbf{(2.1.6.4)}$$

Next we need to compute the point of tangency

$$P = \left(f\left(\frac{5\pi}{4}\right)\cos\left(\frac{5\pi}{4}\right), f\left(\frac{5\pi}{4}\right)\sin\left(\frac{5\pi}{4}\right) \right)$$

```
> P:=expand(subs(theta=5*Pi/4, [f(theta)*cos(theta), f
  (theta)*sin(theta)]));
```

$$P := \left[-\frac{3}{2}\sqrt{2} - \frac{3}{2}, -\frac{3}{2}\sqrt{2} - \frac{3}{2} \right] \qquad \textbf{(2.1.6.5)}$$

Now we can generate the equation of the tangent line.

```
> TL:=y=m*(x-P[1])+P[2];
```

$$TL := y = \left(-\sqrt{2} - 1\right)\left(x + \frac{3}{2}\sqrt{2} + \frac{3}{2}\right) - \frac{3}{2}\sqrt{2} - \frac{3}{2} \qquad \textbf{(2.1.6.6)}$$

```
> TL:=expand(TL);
  evalf(TL);
```

$$TL := y = -\sqrt{2}\,x - 6 - \frac{9}{2}\sqrt{2} - x$$

$$y = -2.414213562\,x - 12.36396103 \qquad \textbf{(2.1.6.7)}$$

Finally, we display the images.

```
> p1:=polarplot(f(theta), theta=0..2*Pi, color=blue):
  p2:=plot(rhs(TL), x=-5..5, color=red):
  display([p1, p2], view=[-5..5, -7..2], scaling=
  constrained);
```




```
>
```